COE 202, Term 203

Digital Logic Design

Assignment# 4 Solution
Due date: Saturday, August 7, 2021

It is required to design a synchronous sequential circuit that receives a serial input X that produces 1 when the input sequence is either {0110} (i.e., 0 followed by 1 followed by 1 followed by 0) or {1100} (i.e., 1 followed by 1 followed by 0 followed by 0) assuming overlapping sequences. The output Z is also a bit stream that produces a 1 only after detecting any of the two sequences. Use an asynchronous reset input to reset the sequential circuit to its initial state.

Example:

a) (10 points) Draw a Mealy state diagram of the sequential circuit.

b) (10 points) Implement your design using a minimal number of D-type flip flops and combinational logic. Specify the K-maps and write the minimal next state and output equations.
We will use the state assignment: 
S0 = 000, S1 = 001, S2 = 010, S3 = 011, S4 = 100, S5 = 101, S6 = 110. 

The K-Maps are as follows:

A+ = A’BX+AB’

B+ = A’B’X+ABX+AB’X’

C+ = A’X’ + BX’ + A’C’

Z = ABX’ + ACX’ 
c) (10 points) Write a structural Verilog model that models your implemented sequential circuit by modeling the D Flip-Flops and instantiating them and modeling the combinational part using either assign statement or gate primitives.
module dff (output reg q, output q_bar, input data, set, rst, clk); 

  assign q_bar = !q; 

  always @(posedge clk, posedge set, posedge rst) 

    if (rst) q <= 0; 

  else if (set) q <=1; else q <= data; 

endmodule

module Ass4_1(output Z, input X, Reset, CLK); 

  wire [2:0] stateS; 

  assign stateS = {A, B, C}; 

  dff F0 (A, Ab, DA, 1'b0, Reset, CLK); 

  dff F1 (B, Bb, DB, 1'b0, Reset, CLK); 

  dff F2 (C, Cb, DC, 1'b0, Reset, CLK); 

  assign DA = ~A & B & X | A & ~B; 

  assign DB = ~A & ~B & X | A & B & X | A & ~B & ~X ; 

  assign DC = ~A & ~X | B & ~X | ~A & ~C;

  assign Z = A & B & ~X | A & C & ~X;

endmodule
d) (10 points) Write a behavioral Verilog description that models your state diagram in part (a).
module Ass4_2 ( output reg Z, input X,Reset, Clk);

  parameter S0 = 3'b000, S1 = 3'b001 , S2 = 3'b010, S3 = 3'b011,S4 = 3'b100, S5 = 3'b101 , S6 = 3'b110;

  reg [2:0] state, N_state; 

  always @( posedge Clk, posedge Reset) begin    

    if (Reset) state <= S0;

    else state <= N_state; 

  end

  always @(X,state) begin 

    Z = 0;   

    case (state) 

      S0 : if(X)  N_state =S3; else N_state = S1;

      S1 : if(X)  N_state =S2; else N_state = S1; 

      S2 : if(X)  N_state =S5; else N_state = S1;

      S3 : if(X)  N_state =S4; else N_state = S1;

      S4 : if(X)  N_state =S4; else N_state = S6;

      S5 : if(X)  N_state =S4; else begin  N_state = S6;Z = 1;end

      S6 : if(X)  N_state =S2; else begin  N_state = S1;Z = 1; end
      default: N_state = 'bx;

   endcase

  end 

endmodule
e) (10 points) Write a test bench that tests BOTH the structural Verilog model of part (c) and the behavioral Verilog model of part (d) using the example input sequence shown above. Call the outputs Z1 and Z2 for the models in (c) and (d), respectively. Start by resetting all flip-flops and then apply the input sequence of X. Verify that your circuit produces the correct output by including the generated waveform from simulation and comparing to the given example above.
module Test_Ass4(); 

  wire Z1, Z2; 

  reg X, rst, clk;

  Ass4_1 M1 (Z1, X, rst, clk);
  Ass4_2 M2 (Z2, X, rst, clk); 

initial begin 

  rst = 0; clk = 0; #10 rst = 1; #10 rst = 0; forever #10 clk = ~ clk; end

initial begin 

    $dumpfile("dump.vcd"); 

    $dumpvars(0, Test_Ass4); 

    X=1; 

    @(negedge clk) X=1; 

    @(negedge clk) X=1; 

    @(negedge clk) X=0; 

    @(negedge clk) X=0; 

    @(negedge clk) X=1; 

    @(negedge clk) X=1;    

    @(negedge clk) X=0; 

    @(negedge clk) X=0; 

    @(negedge clk) X=1;

    @(negedge clk) X=1; 

    @(negedge clk) X=0; 

    @(negedge clk) X=1;
    @(negedge clk) X=0;   

    #20 $finish; 

  end 

endmodule
The simulation waveform is shown below which shows that the correct output sequences are obtained as desired:

f) Submit a report (Word or PDF document) that should contain:

i. The state diagram of your design (part a).

ii. The state table, K-maps, and logic equations of your sequential circuit (part b).

iii. A copy of the Verilog modules and test benches of parts (c), (d), and (e).

iv. The timing diagrams (waveforms) taken directly as snapshots from the simulator for parts

(e) and analysis of output.

The assignment can be solved individually, or in groups of two students. Submit a soft copy of the report on Blackboard.

